
Today

⬛ Review of a few tricky bits from yesterday

⬛ Basics of control flow

⬛ Condition codes

⬛ Conditional operations

⬛ Loops

⬛ Activity: understanding disassembled control flow

⬛ If we have time: switch statements

Reminder: Machine Instructions
⬛C

▪ Store value t where designated by 
dest

⬛Assembly
▪ Move 8-byte value to memory

▪ Quad words in x86-64 parlance

▪ Operands:

t: Register %rax

dest: Register %rbx

*dest: MemoryM[%rbx]

⬛Machine
▪ 3 bytes at address 0x40059e

▪ Compact representation of the 
assembly instruction

▪ (Relatively) easy for hardware to 
interpret

*dest = t;

movq %rax, (%rbx)

0x40059e:  48 89 03

Reminder: Machine Instructions
⬛C

▪ Store value t where designated by 
dest

⬛Assembly
▪ Move 8-byte value to memory

▪ Quad words in x86-64 parlance

▪ Operands:

t: Register %rax

dest: Register %rbx

*dest: MemoryM[%rbx]

⬛Machine
▪ 3 bytes at address 0x40059e

▪ Compact representation of the 
assembly instruction

▪ (Relatively) easy for hardware to 
interpret

*dest = t;

movq %rax, (%rbx)

0x40059e:  48 89 03

0100 1 0 0 0  10001011  00 000 011

REX  W R X B    Move   Mod   R   M

Reminder: Address Modes

⬛Most General Form

D(Rb,Ri,S) Mem[Reg[Rb]+S*Reg[Ri]+ D]
▪ D: Constant “displacement” 1, 2, or 4 bytes

▪ Rb: Base register: Any of 16 integer registers

▪ Ri: Index register: Any, except for %rsp

▪ S: Scale: 1, 2, 4, or 8 (why these numbers?)

⬛Special Cases

(Rb,Ri) Mem[Reg[Rb]+Reg[Ri]]

D(Rb,Ri) Mem[Reg[Rb]+Reg[Ri]+D]

(Rb,Ri,S) Mem[Reg[Rb]+S*Reg[Ri]]

Memory operands and LEA

⬛ In most instructions, a memory operand accesses memory

⬛ LEA is special: it doesn’t access memory

Assembly C equivalent

mov 6(%rbx,%rdi,8), %ax ax = *(rbx + rdi*8 + 6)

add 6(%rbx,%rdi,8), %ax ax += *(rbx + rdi*8 + 6)

xor %ax, 6(%rbx,%rdi,8) *(rbx + rdi*8 + 6) ^= ax

Assembly C equivalent

lea 6(%rbx,%rdi,8), %rax rax = rbx + rdi*8 + 6

Why use LEA?

 CPU designers’ intended use: calculate a pointer to an object

▪ An array element, perhaps

▪ For instance, to pass just one array element to another function

 Compiler authors like to use it for ordinary arithmetic
▪ It can do complex calculations in one instruction

▪ It’s one of the only three-operand instructions the x86 has

▪ It doesn’t touch the condition codes (we’ll come back to this)

Assembly C equivalent

lea (%rbx,%rdi,8), %rax rax = &rbx[rdi]

Assembly C equivalent

lea (%rbx,%rbx,2), %rax rax = rbx * 3

⬛ Basics of control flow

⬛ Condition codes

⬛ Loops

⬛ Activity: understanding disassembled control flow

⬛ If we have time: switch statements

Control flow 

extern void op1(void);

extern void op2(void);

void decision(int x) {

if (x) {

op1();

} else {

op2();

}

}

decision

x != 0

op2 op1

return

Control flow in assembly language

extern void op1(void);

extern void op2(void);

void decision(int x) {

if (x) {

op1();

} else {

op2();

}

}

decision:

subq $8, %rsp

testl %edi, %edi

je      .L2

call op1

jmp .L1

.L2:

call op2

.L1:

addq $8, %rsp

ret

It’s all done with 
GOTO!

Processor State (x86-64, Partial)

⬛Information about 
currently executing 
program

▪ Temporary data
( %rax, … )

▪ Location of runtime stack
( %rsp )

▪ Location of current code 
control point
( %rip, … )

▪ Status of recent tests
( CF, ZF, SF, OF )

%rip

Registers

Current stack top
Instruction pointer

CF ZF SF OF Condition codes

%rsp

%r8

%r9

%r10

%r11

%r12

%r13

%r14

%r15

%rax

%rbx

%rcx

%rdx

%rsi

%rdi

%rbp

Condition Codes (Implicit Setting)

⬛Single bit registers

▪ CF Carry Flag (for unsigned) SF Sign Flag (for signed)

▪ ZF Zero Flag OF Overflow Flag (for signed)

⬛Implicitly set (as side effect) of arithmetic operations
Example: addq Src,Dest↔ t = a+b

CF set if carry out from most significant bit (unsigned overflow)

ZF set if t == 0

SF set if t < 0 (as signed)

OF set if two’s-complement (signed) overflow
(a>0 && b>0 && t<0) || (a<0 && b<0 && t>=0)

⬛Not set by leaq instruction

ZF set when

000000000000…00000000000

SF set when

1xxxxxxxxxxx…xxxxxxxxxxx

CF set when

yxxxxxxxxxxxx...

yxxxxxxxxxxxx...+

zxxxxxxxxxxxx...1

OF set when

yxxxxxxxxxxxx...

yxxxxxxxxxxxx...+

zxxxxxxxxxxxx...

z != y

Compare Instruction

⬛ cmp a, b

▪ Computes 𝑏 − 𝑎 (just like sub)

▪ Sets condition codes based on result, but…

▪ Does not change 𝒃

▪ Used for if (a < b) { … }
whenever 𝑏 − 𝑎 isn’t needed for anything else

Test Instruction

⬛ test a, b

▪ Computes 𝑏 ∧ 𝑎 (just like and)

▪ Sets condition codes (only SF and ZF) based on result, but…

▪ Does not change 𝒃

▪ Most common use: test %rX, %rX
to compare %rX to zero

▪ Second most common use: test %rX, %rY
tests if any of the 1-bits in %rY are also 1 in %rX (or vice versa)

Reading Condition Codes

⬛SetX Instructions

▪ Set low-order byte of destination to 0 or 1 based on combinations of 
condition codes

▪ Does not alter remaining 7 bytes

SetX Condition Description

sete ZF Equal / Zero

setne ~ZF Not Equal / Not Zero

sets SF Negative

setns ~SF Nonnegative

setg ~(SF^OF)&~ZF Greater (Signed)

setge ~(SF^OF) Greater or Equal (Signed)

setl (SF^OF) Less (Signed)

setle (SF^OF)|ZF Less or Equal (Signed)

seta ~CF&~ZF Above (unsigned)

setb CF Below (unsigned)

%rsp

x86-64 Integer Registers

▪ Can reference low-order byte

%al

%bl

%cl

%dl

%sil

%dil

%spl

%bpl

%r8b

%r9b

%r10b

%r11b

%r12b

%r13b

%r14b

%r15b

%r8

%r9

%r10

%r11

%r12

%r13

%r14

%r15

%rax

%rbx

%rcx

%rdx

%rsi

%rdi

%rbp

cmpq   %rsi, %rdi   # Compare x:y

setg   %al          # Set when >

movzbl %al, %eax    # Zero rest of %rax

ret

Reading Condition Codes (Cont.)
⬛SetX Instructions: 

▪ Set single byte based on combination of condition 
codes

⬛One of addressable byte registers
▪ Does not alter remaining bytes

▪ Typically use movzbl to finish job

▪ 32-bit instructions also set upper 32 bits to 0

int gt (long x, long y)

{

return x > y;

}

Register Use(s)

%rdi Argument x

%rsi Argument y

%rax Return value

cmpq   %rsi, %rdi   # Compare x:y

setg   %al          # Set when >

movzbl %al, %eax    # Zero rest of %rax

ret

Reading Condition Codes (Cont.)
⬛SetX Instructions: 

▪ Set single byte based on combination of condition 
codes

⬛One of addressable byte registers
▪ Does not alter remaining bytes

▪ Typically use movzbl to finish job

▪ 32-bit instructions also set upper 32 bits to 0

int gt (long x, long y)

{

return x > y;

}

Register Use(s)

%rdi Argument x

%rsi Argument y

%rax Return value

Beware weirdness movzbl (and others)

movzbl %al, %eax

%eax %al%rax 0x000000 %al%rax0x00000000 0x000000 %al

Zapped to all 
0’s

Jumping

⬛jX Instructions

▪ Jump to different part of code depending on condition codes

jX Condition Description

jmp 1 Unconditional

je ZF Equal / Zero

jne ~ZF Not Equal / Not Zero

js SF Negative

jns ~SF Nonnegative

jg ~(SF^OF)&~ZF Greater (Signed)

jge ~(SF^OF) Greater or Equal (Signed)

jl (SF^OF) Less (Signed)

jle (SF^OF)|ZF Less or Equal (Signed)

ja ~CF&~ZF Above (unsigned)

jb CF Below (unsigned)

Conditional Branch Example (Old Style)

long absdiff

(long x, long y)

{

long result;

if (x > y)

result = x-y;

else

result = y-x;

return result;

}

absdiff:

cmpq    %rsi, %rdi  # x:y

jle     .L4

movq    %rdi, %rax

subq    %rsi, %rax

ret

.L4:       # x <= y

movq    %rsi, %rax

subq    %rdi, %rax

ret

⬛ Generation
shark> gcc –Og -S –fno-if-conversion control.c

Register Use(s)

%rdi Argument x

%rsi Argument y

%rax Return value

I’ll get to this 
shortly.

Expressing with Goto Code

long absdiff

(long x, long y)

{

long result;

if (x > y)

result = x-y;

else

result = y-x;

return result;

}

⬛ C allows goto statement

⬛ Jump to position designated by label

long absdiff_j

(long x, long y)

{

long result;

int ntest = x <= y;

if (ntest) goto Else;

result = x-y;

goto Done;

Else:

result = y-x;

Done:

return result;

}

C Code

val = Test ? Then_Expr : Else_Expr;

Goto Version
ntest = !Test;
if (ntest) goto 

Else;

val = Then_Expr;
goto Done;

Else:

val = Else_Expr;
Done:

. . .

General Conditional Expression Translation 
(Using Branches)

▪ Create separate code regions for 
then & else expressions

▪ Execute appropriate one

val = x>y ? x-y : y-x;

C Code

val = Test
? Then_Expr
: Else_Expr;

Goto Version

result = Then_Expr;
eval = Else_Expr;
nt = !Test;
if (nt) result = eval;

return result;

Using Conditional Moves
⬛Conditional Move Instructions

▪ Instruction supports:

if (Test) Dest ← Src

▪ Supported in post-1995 x86 processors

▪ GCC tries to use them

▪ But, only when known to be safe

⬛Why?
▪ Branches are very disruptive to 

instruction flow through pipelines

▪ Conditional moves do not require 
control transfer

Conditional Move Example

absdiff:

movq    %rdi, %rax  # x

subq    %rsi, %rax  # result = x-y

movq    %rsi, %rdx

subq    %rdi, %rdx  # eval = y-x

cmpq    %rsi, %rdi  # x:y

cmovle  %rdx, %rax  # if <=, result = eval

ret

long absdiff

(long x, long y)

{

long result;

if (x > y)

result = x-y;

else

result = y-x;

return result;

}

Register Use(s)

%rdi Argument x

%rsi Argument y

%rax Return value

Expensive Computations

Bad Cases for Conditional Move

⬛ Both values get computed

⬛Only makes sense when computations 
are very simple

val = Test(x) ? Hard1(x) : Hard2(x);

Risky Computations

⬛ Both values get computed

⬛May have undesirable effects

val = p ? *p : 0;

Computations with side effects

⬛ Both values get computed

⬛Must be side-effect free

val = x > 0 ? x*=7 : x+=3;

Bad Performance

Unsafe

Illegal

xor    %rax, %rax

sub    $1, %rax 

cmp    $2, %rax

setl   %al

movzbl %al, %eax

%rax SF CF OF ZF

0x0000 0000 0000 0000 0 0 0 1

0xFFFF FFFF FFFF FFFF 1 1 0 0

0xFFFF FFFF FFFF FFFF 1 0 0 0

0xFFFF FFFF FFFF FF01 1 0 0 0

0x0000 0000 0000 0001 1 0 0 0

Exercise

xor    %rax, %rax

sub    $1, %rax 

cmp    $2, %rax

setl   %al

movzbl %al, %eax

%rax SF CF OF ZF

0x0000 0000 0000 0000 0 0 0 1

0xFFFF FFFF FFFF FFFF 1 1 0 0

0xFFFF FFFF FFFF FFFF 1 0 0 0

0xFFFF FFFF FFFF FF01 1 0 0 0

0x0000 0000 0000 0001 1 0 0 0

Exercise

⬛ Basics of control flow

⬛ Condition codes

⬛ Conditional operations

⬛ Loops

⬛ Activity: understanding disassembled control flow

⬛ If we have time: switch statements

C Code
long pcount_do

(unsigned long x) {

long result = 0;

do {

result += x & 0x1;

x >>= 1;

} while (x);

return result;

}

Goto Version
long pcount_goto

(unsigned long x) {

long result = 0;

loop:

result += x & 0x1;

x >>= 1;

if(x) goto loop;

return result;

}

“Do-While” Loop Example

⬛Count number of 1’s in argument x (“popcount”)

⬛Use conditional branch to either continue looping or to exit 
loop

Goto Version

“Do-While” Loop Compilation

movl $0, %eax #  

result = 0

.L2: # loop:

movq %rdi, %rdx

andl $1, %edx #  t = x & 0x1

addq %rdx, %rax #  result += t

shrq %rdi #  x >>= 1

jne .L2 #  if (x) goto

loop

rep; ret

long pcount_goto

(unsigned long x) {

long result = 0;

loop:

result += x & 0x1;

x >>= 1;

if(x) goto loop;

return result;

}

Register Use(s)

%rdi Argument x

%rax result



36Bryant and O’Hallaron, Computer Systems: A Programmer’s Perspective, Third Edition

Carnegie Mellon

C Code

do 

Body
while (Test);

Goto Version

loop:

Body
if (Test)
goto loop

General “Do-While” Translation

⬛Body: {

Statement1;

Statement2;

…

Statementn;

}

While version

while (Test)
Body

General “While” Translation #1

⬛ “Jump-to-middle” translation

⬛ Used with -Og Goto Version

goto test;

loop:

Body
test:

if (Test)
goto loop;

done:

C Code
long pcount_while

(unsigned long x) {

long result = 0;

while (x) {

result += x & 0x1;

x >>= 1;

}

return result;

}

Jump to Middle 
Versionlong pcount_goto_jtm

(unsigned long x) {

long result = 0;

goto test;

loop:

result += x & 0x1;

x >>= 1;

test:

if(x) goto loop;

return result;

}

While Loop Example #1

⬛Compare to do-while version of function

⬛Initial goto starts loop at test

While version

while (Test)
Body

Do-While Version

if (!Test) 
goto done;

do

Body
while(Test);

done:

General “While” Translation #2

⬛ “Do-while” conversion

⬛ Used with –O1

Goto Version

if (!Test)
goto done;

loop:

Body
if (Test)

goto loop;

done:

C Code
long pcount_while

(unsigned long x) {

long result = 0;

while (x) {

result += x & 0x1;

x >>= 1;

}

return result;

}

Do-While Version
long pcount_goto_dw

(unsigned long x) {

long result = 0;

if (!x) goto done;

loop:

result += x & 0x1;

x >>= 1;

if(x) goto loop;

done:

return result;

}

While Loop Example #2

⬛ Compare to do-while version of function

⬛ Initial conditional guards entrance to loop

“For” Loop Form

for (Init; Test; Update )

Body

General Form

#define WSIZE 8*sizeof(int)

long pcount_for

(unsigned long x)

{

size_t i;

long result = 0;

for (i = 0; i < WSIZE; i++)

{

unsigned bit = 

(x >> i) & 0x1;

result += bit;

}

return result;

}

i = 0

i < WSIZE

i++

{

unsigned bit =

(x >> i) & 0x1;

result += bit;

}

Init

Test

Update

Body

“For” Loop → While Loop

for (Init; Test; Update )

Body

For Version

Init;

while (Test ) {

Body

Update;

}

While Version



43Bryant and O’Hallaron, Computer Systems: A Programmer’s Perspective, Third Edition

Carnegie Mellon

For-While Conversion
long pcount_for_while

(unsigned long x)

{

size_t i;

long result = 0;

i = 0;

while (i < WSIZE)

{

unsigned bit = 

(x >> i) & 0x1;

result += bit;

i++;

}

return result;

}

i = 0

i < WSIZE

i++

{

unsigned bit =

(x >> i) & 0x1;

result += bit;

}

Init

Test

Update

Body

C Code

“For” Loop Do-While Conversion

⬛ Initial test can be optimized 
away

long pcount_for

(unsigned long x)

{

size_t i;

long result = 0;

for (i = 0; i < WSIZE; i++)

{

unsigned bit = 

(x >> i) & 0x1;

result += bit;

}

return result;

}

Goto Version
long pcount_for_goto_dw

(unsigned long x) {

size_t i;

long result = 0;

i = 0;

if (!(i < WSIZE))

goto done;

loop:

{

unsigned bit = 

(x >> i) & 0x1;

result += bit;

}

i++;

if (i < WSIZE)

goto loop;

done:

return result;

}

Ini
t
!Test

Body

Update

Test

Today

⬛ Review of a few tricky bits from yesterday

⬛ Basics of control flow

⬛ Condition codes

⬛ Conditional operations

⬛ Loops

⬛ Activity: understanding disassembled control flow

⬛ If we have time: switch statements

https://www.cs.cmu.edu/~213/activities/213_lecture6.pdf

https://www.cs.cmu.edu/~213/activities/213_lecture6.pdf

⬛ Condition codes

⬛ Loops

⬛ Activity: understanding disassembled control flow

⬛ If we have time: switch statements

Switch Statement 
Example

⬛ Multiple case labels

▪ Here: 5 & 6

⬛ Fall through cases
▪ Here: 2

⬛ Missing cases
▪ Here: 4

long switch_eg

(long x, long y, long z)

{

long w = 1;

switch(x) {

case 1:

w = y*z;

break;

case 2:

w = y/z;

/* Fall Through */

case 3:

w += z;

break;

case 5:

case 6:

w -= z;

break;

default:

w = 2;

}

return w;

}

Jump Table Structure

Code Block
0

Targ0:

Code Block
1

Targ1:

Code Block
2

Targ2:

Code Block
n–1

Targn-1:

•

•

•

Targ0

Targ1

Targ2

Targn-1

•

•

•

jtab:

goto *JTab[x];

switch(x) {

case val_0:

Block 0
case val_1:

Block 1
• • •

case val_n-1:

Block n–1
}

Switch Form

Translation (Extended 
C)

Jump 
Table

Jump 
Targets

Switch Statement Example

Setup:

long switch_eg(long x, long y, long z)

{

long w = 1;

switch(x) {

. . .

}

return w;

}

switch_eg:

movq    %rdx, %rcx

cmpq    $6, %rdi   # x:6

ja      .L8

jmp     *.L4(,%rdi,8)

What range of values 
takes default?

Note that w not 
initialized here

Register Use(s)

%rdi Argument x

%rsi Argument y

%rdx Argument z

%rax Return value

Switch Statement Example

long switch_eg(long x, long y, long z)

{

long w = 1;

switch(x) {

. . .

}

return w;

}

Indirect 
jump

Jump table
.section .rodata

.align 8

.L4:

.quad .L8

# x = 0

.quad .L3

# x = 1

.quad .L5

# x = 2

.quad .L9

# x = 3

.quad .L8

# x = 4

.quad .L7

# x = 5

.quad .L7

# x = 6

Setup:

switch_eg:

movq    %rdx, %rcx

cmpq    $6, %rdi      # x:6

ja      .L8           # Use default

jmp     *.L4(,%rdi,8) # goto *JTab[x]

Assembly Setup Explanation

⬛ Table Structure

▪ Each target requires 8 bytes

▪ Base address at .L4

⬛ Jumping
▪ Direct: jmp .L8

▪ Jump target is denoted by label .L8

▪ Indirect: jmp *.L4(,%rdi,8)

▪ Start of jump table: .L4

▪ Must scale by factor of 8 (addresses are 8 bytes)

▪ Fetch target from effective Address .L4 + x*8

▪ Only for  0 ≤ x ≤ 6

Jump table

.section .rodata

.align 8

.L4:

.quad .L8

# x = 0

.quad .L3

# x = 1

.quad .L5

# x = 2

.quad .L9

# x = 3

.quad .L8

# x = 4

.quad .L7

# x = 5

.quad .L7

# x = 6

.section .rodata

.align 8

.L4:

.quad .L8

# x = 0

.quad .L3

# x = 1

.quad .L5

# x = 2

.quad .L9

# x = 3

.quad .L8

# x = 4

.quad .L7

# x = 5

.quad .L7

# x = 6

Jump Table

Jump table
switch(x) {

case 1:      // .L3

w = y*z;

break;

case 2:      // .L5

w = y/z;

/* Fall Through */

case 3:      // .L9

w += z;

break;

case 5:

case 6:      // .L7

w -= z;

break;

default:     // .L8

w = 2;

}

Code Blocks (x == 1)
.L3:

movq    %rsi, %rax  # y

imulq   %rdx, %rax  # y*z

ret

switch(x) {

case 1: // .L3

w = y*z;

break;

. . .

}

Register Use(s)

%rdi Argument x

%rsi Argument y

%rdx Argument z

%rax Return value

Handling Fall-Through

long w = 1;

. . .

switch(x) {

. . .

case 2:

w = y/z;

/* Fall Through */

case 3:

w += z;

break;

. . .

}
case 3:

w = 1;

case 2:

w = y/z;

goto merge;

merge:

w += z;

Code Blocks (x == 2, x == 3)
.L5:                  # Case 2

movq    %rsi, %rax

cqto

idivq   %rcx       #  y/z

jmp     .L6        #  goto merge

.L9:                  # Case 3

movl    $1, %eax   #  w = 1

.L6:                  # merge:

addq    %rcx, %rax #  w += z

ret

long w = 1;

. . .

switch(x) {

. . .

case 2:

w = y/z;

/* Fall Through */

case 3:

w += z;

break;

. . .

} Register Use(s)

%rdi Argument x

%rsi Argument y

%rdx Argument z

%rax Return value



56Bryant and O’Hallaron, Computer Systems: A Programmer’s Perspective, Third Edition

Carnegie Mellon

Code Blocks (x == 5, x == 6, default)
.L7:               # Case 5,6

movl  $1, %eax   #  w = 1

subq  %rdx, %rax #  w -= z

ret

.L8:               # Default:

movl  $2, %eax   #  2

ret

switch(x) {

. . .

case 5:  // .L7

case 6:  // .L7

w -= z;

break;

default: // .L8

w = 2; 

}

Register Use(s)

%rdi Argument x

%rsi Argument y

%rdx Argument z

%rax Return value



57Bryant and O’Hallaron, Computer Systems: A Programmer’s Perspective, Third Edition

Carnegie Mellon

Finding Jump Table in Binary

00000000004005e0 <switch_eg>:

4005e0:       48 89 d1                mov    %rdx,%rcx

4005e3:       48 83 ff 06             cmp $0x6,%rdi

4005e7:       77 2b                   ja     400614 <switch_eg+0x34>

4005e9:       ff 24 fd f0 07 40 00    jmpq *0x4007f0(,%rdi,8)

4005f0:       48 89 f0                mov    %rsi,%rax

4005f3:       48 0f af c2             imul %rdx,%rax

4005f7:       c3                      retq

4005f8:       48 89 f0                mov    %rsi,%rax

4005fb:       48 99                   cqto

4005fd:       48 f7 f9                idiv %rcx

400600:       eb 05                   jmp 400607 <switch_eg+0x27>

400602:       b8 01 00 00 00          mov    $0x1,%eax

400607:       48 01 c8                add    %rcx,%rax

40060a:       c3                      retq

40060b:       b8 01 00 00 00          mov    $0x1,%eax

400610:       48 29 d0                sub    %rdx,%rax

400613:       c3                      retq

400614:       b8 02 00 00 00          mov    $0x2,%eax

400619:       c3                      retq

Finding Jump Table in Binary (cont.)

00000000004005e0 <switch_eg>:

. . .

4005e9:       ff 24 fd f0 07 40 00    jmpq *0x4007f0(,%rdi,8)

. . .

% gdb switch

(gdb) x /8xg 0x4007f0

0x4007f0:       0x0000000000400614      0x00000000004005f0

0x400800:       0x00000000004005f8      0x0000000000400602

0x400810:       0x0000000000400614      0x000000000040060b

0x400820:       0x000000000040060b      0x2c646c25203d2078

(gdb) 

Finding Jump Table in Binary (cont.)
% gdb switch

(gdb) x /8xg 0x4007f0

0x4007f0:       0x0000000000400614      0x00000000004005f0

0x400800:       0x00000000004005f8      0x0000000000400602

0x400810:       0x0000000000400614      0x000000000040060b

0x400820:       0x000000000040060b      0x2c646c25203d2078

. . .

4005f0:       48 89 f0                mov    %rsi,%rax

4005f3:       48 0f af c2             imul %rdx,%rax

4005f7:       c3                      retq

4005f8:       48 89 f0                mov    %rsi,%rax

4005fb:       48 99                   cqto

4005fd:       48 f7 f9                idiv %rcx

400600:       eb 05                   jmp 400607 <switch_eg+0x27>

400602:       b8 01 00 00 00          mov    $0x1,%eax

400607:       48 01 c8                add    %rcx,%rax

40060a:       c3                      retq

40060b:       b8 01 00 00 00          mov    $0x1,%eax

400610:       48 29 d0                sub    %rdx,%rax

400613:       c3                      retq

400614:       b8 02 00 00 00          mov    $0x2,%eax

400619:       c3                      retq


